THE UNIVERSITY OF WISCONSIN—MILWAUKEE

CENTER
FOR
GREAT LAKES STUDIES

MILWAUKEE, WISCONSIN 53201 U.S.A.

SPECIAL REPORT NO. 4

Digital Computer Programs for the Defant Method of Seiche Analysis

by

EVERETT J. FEE

Center for Great Lakes Studies

The University of Wisconsin-Milwaukee
Milwaukee, Wisconsin 53201 USA

July, 1968

CONTENTS

LIST OF FIGURES

DIGITAL COMPUTER PROGRAMS FOR THE DEFANT METHOD OF SEICHE ANALYSIS

INTRODUCTION

Surface seiches are long, standing waves which can occur in closed basins or bays. To compute the seiche parameters in actual basins, gulfs or bays, several procedures are available. Of these, the Defant (1961) method of analysis is the most useful. However, the lack of readily available computer programs has limited the scope of its application, because of the burden of reprogramming. The programs presented here should eliminate much duplicated effort.

Seiches are initiated when a force acts impulsively over a large part of the waterbody. Steady wind stress can also start a seiche by inducing a temporary displacement of part of the water, so that at equilibrium the acceleration due to the component of gravity acting down the slope of the water surface balances the stress. When the wind stress is then removed, the displaced water moves back toward the position of static equilibrium, but over-shoots because of its kinetic energy. The oscillation thus initiated continues until altered by another external force or until the energy is lost by friction.

Figure 1 shows the horizontal and vertical water displacements associated with the simplest of this class of waves. Any harmonic of the basic mode may be generated, but the lower harmonics usually dominate because of the greater frictional damping per unit time of the higher modes (Mortimer, 1953). Since seiches are free waves, their periods are determined by the morphometry of the basin. The magnitude of the response of the lake to any applied force depends on the magnitude of the force and whether the forcing function is near resonance with any of the free periods of the lake (Mortimer, 1965).

A number of computational schemes have been developed for predicting the periods of seiches in real lakes. Hutchinson (1957) and Defant (1961) have given thorough reviews of these techniques. The Defant (1918; in English, 1961) method has considerable advantages over most of the other methods in that it gives the relative horizontal and vertical displacements of water particles associated with the seiche in the same computation as that giving the period. The limitations of the method are known both empirically (Fee and Bachmann, 1968) and theoretically, so the amount of effort expended can be adjusted to the accuracy desired. The basic technique may also be used for the solution of other problems, for example in computing the phase speeds of internal waves for general vertical density profiles (Johnson and Fee, 1968).

This report describes the theory and applications of the Defant method along with appropriate computer programs. It is hoped that this will allow a wider use of this unique tool.

THEORY

The basic equations describing water motion in a seiche can be explained using Figure 2. The symbols used in the formulae are listed in Table 1.

TABLE 1. Symbols used in the text and figures

x and X	a point along the axis of the lake or bay
ξ_x	the horizontal displacement of water particles at x
η_x	the vertical displacement of water particles at x
S_x	the area of a cross-section of the lake at the point x
q_x	the volume of water passing through a cross-section at x
v_x	the surface area of the lake or bay between sections at x and x-1
b_x	the breadth of the lake or bay at the point x
g	the acceleration due to gravity
ℓ	the length of the "Talweg" (defined in the text)
t	time
h	the mean depth of the lake or bay
Δ	the difference operator
T	the period of free oscillation of the lake or bay

The equation of continuity is an expression of the conservation of matter and may be derived as follows. The difference between the volume of water passing through sections X_1 and X_2 (Fig. 2) is seen to be approximately the area of the surface of the lake times the mean vertical displacement between the two sections.

$$\Delta q = \left[(b_1 + b_2)/2\right] \Delta X_2 \left[(\eta_1 + \eta_2)/2\right] = v_2(\eta_1 + \eta_2)/2$$

This must equal the change in volume of the compartment of the lake between X_1 and X_2, which is $(S_1\xi_1 - S_2\xi_2)$. Thus, the finite difference form of the equation of continuity is

$$-\Delta(S\xi) \cong \bar{\eta}\,\bar{b}\,\Delta x$$

where the bars indicate averages. As Δx approaches zero, this becomes the equation of continuity

$$\frac{\partial(S\xi)}{\partial x} = -\eta b \tag{1}$$

The equation of motion simply states that the acceleration of a water particle is proportional to the slope of the free surface above it. This is formulated as

$$\frac{\partial^2 \xi}{\partial t^2} = -g \frac{\partial \eta}{\partial x} \tag{2}$$

The Defant analysis starts with equations (1) and (2). The horizontal displacement in a seiche is assumed to be a simple harmonic function of time and an unspecified function of position,

$$\xi = \xi_o(x) \cos(2\pi t/T)$$

Taking two derivatives of ξ gives

$$\frac{\partial^2 \xi}{\partial t^2} = -(4\pi^2/T^2)\,\xi$$

It follows from (2) that

$$\partial\eta = (4\pi^2/gT^2)\,\xi\,\partial x$$

Introducing finite differentials and letting

$$\alpha = (4\pi^2/gT^2)\,\Delta x$$

gives

$$\eta_2 = \eta_1 + \alpha \xi$$

Assuming linearity in ξ, one obtains

$$\eta_2 = \eta_1 + (\xi_1 + \xi_2)(\alpha/2) \tag{3}$$

where subscripts $_1$ and $_2$ refer to successive cross sections.

It was previously shown that the volume increment of the lake between sections $_1$ and $_2$ is approximately $(v_2(\eta_1 + \eta_2)/2)$. Thus, the total volume passing through section $_2$ is

$$q_2 = q_1 + \Delta q \quad \text{where } \Delta q = v_2(\eta_1 + \eta_2)/2 \tag{4}$$

Equations (3) and (4) are now used to derive an analogous relation for ξ. Integration of (1) gives

$$\xi_X = \frac{-1}{S_X} \int_0^X \eta(x)\, b(x)\, dx$$

where x is a dummy variable. Now $S_X \xi_X$ is the volume of water passing through the section at x. Thus

$$\xi_2 = -\frac{1}{S_2}\{q_1 + \Delta q\} = -\frac{1}{S_2}\{q_1 + [v_2 + (\eta_1 + \alpha\xi_1/4)/2]\}$$

Using (3) and solving for ξ_2 gives

$$\xi_2 = \frac{-1}{(S_2 + \alpha v_2/4)} [q_1 + (\eta_1 + \alpha\xi_1/4) v_2/2] \tag{5}$$

Equations (3), (4), and (5) are the equations used in the Defant analysis.

PROCEDURE

Starting with the best available chart of the lake or bay, the initial and most subjective step in applying the Defant method is to draw a straight line or a smooth curve along the deeper parts of the basin. Following Defant, this line is called the "Talweg" and represents the assumed track of the standing wave. A number of cross-sections are then drawn approximately perpendicular to the Talweg. It is not necessary that the cross-sections be equally spaced. Indeed, best results are usually obtained if they are rather widely spaced where the basin is regular and closely spaced where it changes rapidly. The number of cross-sections needed has been investigated by Fee and Bachmann (1968). Four lakes, ranging from the very complex basin of Lake West Okoboji in northeast Iowa (Bachmann, et al., 1966) to a theoretical uniform rectangular basin, were considered. In all basins, with only 10 to 20 sections, the Defant method gave an answer for the uninodal seiche 99% of that given by 160 sections (88 sections in the case of Lake Tahoe). The positions of the nodes were unaffected by the number of sections. The vertical displacements, however, were very sensitive to the number of sections and upwards of 80 were needed to define the wave profile in asymetrical basins. It appears that in most cases twenty sections will be sufficient to give the period and the positions of the nodes of the uninodal seiche. A proportionate increase would be necessary for finding the parameters of the higher harmonics.

After determining the positions of the cross-sections, the areas of the cross-sections and the surface areas of the lake between successive cross-sections are measured. These data together with the distances between cross-sections are used with equations (3), (4), and (5) to carry out the numerical integration.

The calculations are started by specifying that $\eta_1 = 100$, $q_1 = 0$, and $\xi_1 = 0$ at one end of the lake. An approximation to the correct period is made and is used to compute α. Merian's formula [$T = 2\,\ell / \sqrt{gh}$] is normally used for this first approximation. With this information, ξ_2 is computed using equation (5), and is in turn used in equation (3) to find η_2. Knowing η_2, equation (4) can be used to get q_2. These three numbers are then used to compute the same parameters for the next section. If the approximation to the period was correct, ξ_2 will be zero when the other end of the lake is reached. Normally many values for the period must be tested before an adequate answer can be obtained. The calculations become tedious but are readily adapted to computer solution. Figure 3 shows the logic used in programming the problem for a digital computer. Appendix 1 is a FORTRAN II program incorporating the logic of Figure 3. The method is easily extended to the study of bay oscillations by specifying that there be a node at the mouth of the bay. Appendix 2 is the FORTRAN program that has been developed for this. This program contains no mouth correction.

The technique used for finding the roots is commonly referred to as the "method of false position" or the "binary chopper." This method does not converge as rapidly as others, for example the Newton-Raphson, but it has the advantage of being independent of the slope of the curve and converging with certainty in a specified number of iterations. Other methods are highly sensitive to the initial approximation to the period and may not converge at all if a poor approximation is used. In practice, the method presented here converges quite quickly. For example on the IBM 360/50 only 3.85 seconds of execution time were required to find the parameters of the uninodal seiche in Lake Tahoe with 8 sections.

SUBROUTINE DEFANT is invoked with a FORTRAN CALL statement. Sufficient documentation is presented in the appendices to explain the data required. Care should be taken that the units of the data coincide with the units used for the acceleration due to gravity. A typical program that would use SUBROUTINE DEFANT is given in Appendix 3. Appendix 4 is an ALGOL program equivalent to the FORTRAN contained in appendices 1 and 3. Figure 3 should allow easy adaptation of this subroutine to other computer languages.



FIGURE 1. Vertical (unbroken line) and horizontal (dotted line) displacements associated with uninodal (upper diagram) and binodal (lower diagram) seiches in a rectangular basin of constant depth.
A_i is the i^{th} antinode of the seiche.
N_i is the i^{th} node of the seiche.



Figure 2. Diagram of a lake illustrating the terminology used.



FIGURE 3. Flow chart of program logic for the Defant analysis for closed basins.



23
ω
α
23
ω
= 0
= 0
12
T₃ ← T₃ - δ
> 0
13
A*ξ₂(N)
= 0
K - 1
> 0
18
B*ξ₂(N)
>0
22
T₁ ← T₃
< 0
< 0
< 0
16
T₁ ← T₃
T₂ ← T₃ + δ
≥0
14
δ
19
T₂ ← T₃
20
|T₃ - (T₁ + T₂)/2| - 0.0001
≤ 0
< 0
> 0
17
K ← 2
T₃ ← (T₁+T₂)/2.
15
T₁ ← T₃ + δ
T₂ ← T₃
8
A ← ξ₂(N)
21
T₃ ← (T₁+T₂)/2.
5
ε
>0
10
<0
A
=0
Nodes - J
≠0
9
A
<0
>0
=0
=0
23
ω
11
δ ← - δ
12
K ← 1
T₃ ← T₃ - δ
23
ω
T₃,
(I, ξ₂(I), η₂(I)
q₂(I), I = 1,N)
RETURN

APPENDIX 1

```
C      DEFANT PROGRAM FOR COMPUTING THE PERIODS OF FREE OSCILLATION
C            OF CLOSED BASINS
C      FORTRAN II
       SUBROUTINE DEFANT(S2, V2, N, DELX, T3, AINCR, IN)
       DIMENSION S2(1), V2(1), HD2(199), VD2(199), Q2(199), DELX(1)
C      S2 IS THE ARRAY OF CROSS SECTION AREAS
C      V2 IS THE ARRAY OF SURFACE AREAS
C      N IS THE NUMBER OF SECTIONS IN THE LAKE
C      DELX IS THE ARRAY OF DISTANCES BETWEEN SUCCESSIVE CROSS SECTIONS
C      T3 IS A CLOSE APPROXIMATION TO THE PERIOD OF OSCILLATION
C      AINCR IS THE INCREMENTING CONSTANT
C      IN IS THE NUMBER OF NODES IN THE SEICHE
       PRINT 1  , IN
     1 FORMAT (1H1, 27HDEFANT CALCULATIONS FOR THE, 2X, I1, 2X, 12H MODAL SEICH
      1E)
       J = IN/2
       J = 2*J
       IF (IN-J)  3, 2, 3
     2 B = -1.
       GO TO 4
     3 B = 1.0
     4 G = 35.29728
C      G IS THE ACCELERATION DUE TO GRAVITY AT THE LATITUDE OF THE LAKE
C      THE UNITS USED HERE ARE KM AND MIN
C      THE UNITS OF THIS CONSTANT DETERMINE THE UNITS OF THE OUTPUT
       IC = 0
C      IC IS A COUNTER FOR THE NUMBER OF ITERATIONS
       K = 0
C      THE BOUNDARY CONDITIONS FOLLOW
     5 VD1 = 100.0
       HD1  = 0.0
```

```
      Q1 = 0.0
      DO 6  I=1,N
      ALPHA = F*DELX(I)/(T3*T3)
      HD2(I) = -(Q1+(VD1 + (ALPHA*HD1)/4.0)*V2(I))/(S2(I)+ALPHA*V2(I)/4.0)
      VD2(I) = VD1+ALPHA*(HD1+HD2(I))/2.0
      Q2(I) = Q1+V2(I)*(VD1+VD2(I))/2.0
      Q1 = Q2(I)
      VD1 = VD2(I)
    6 HD1 = HD2(I)
      IC = IC+1
      PRINT 7  ,IC,T3,HD1,VD1,Q1
    7 FORMAT(1H0,2X,9HITERATION  ,I4,5X,5HT3 =  ,F11.6,5X,6HHD1 =  ,E15.8,5
     1X,6HVD1 =  ,E15.8,5X,5HQ1 =  ,E15.8)
      IF (K-1)  8,13,18
    8 A = HD2(N)
      IF (IN-J)  9,10,9
    9 IF (A)  12,23,11
   10 IF (A)  11,23,12
   11 AINCR = -AINCR
   12 K = 1
      T3 = T3-AINCR
      GO TO 5
   13 IF (A*HD2(N))  14,23,12
   14 IF (AINCR)  15,16,16
   15 T1 = T3+AINCR
      T2 = T3
      GO TO 17
   16 T1 = T3
```

```
      T2 = T3+AINCR
   17 K =2
      GO TO 21
   18 IF (HD2(N)*B)  19,23,22
   19 T2 = T3
   20 IF   (ABS(T3-(T1+T2)/2.0)-0.0001) 23,23,21
   21 T3 =  (T1+T2)/2.0
      GO TO 5
   22 T1 = T3
      GO TO 20
C     CONVERT THE PERIOD TO MORE CONVENIENT TIME UNITS
   23 T =T3/60.0
C     WRITE OUT THE RESULTS
      PRINT 24  ,T,(I,HD2(I),VD2(I),Q2(I),I=1,N)
   24 FORMAT(1H1,37HTHE COMPUTED PERIOD OF OSCILLATION IS  ,2X,F10.4,2X,5
     1HHOURS      //(5X,I3,5X,9HHD2(I) = ,E15.7,5X,9HVD2(I) = ,E15.7,5X,8HQ
     22(I) = ,E15.7))
      RETURN
      END
```

APPENDIX 2

```
C       DEFANT PROGRAM FOR COMPUTING THE PERIODS OF FREE OSCILLATION
C               OF BAYS
C       FORTRAN II
        SUBROUTINE DEFANT(S2,V2,N,DELX,T3,AINCR,IN)
        DIMENSION S2(1),V2(1),HD2(199),VD2(199),Q2(199),DELX(1),X(199)
C       DELX IS THE DISTANCE BETWEEN ADJACENT SECTIONS
C       IN IS THE NUMBER OF NODES IN THE SEICHE
C       AINCR IS THE INCREMENTING CONSTANT
C       N IS THE NUMBER OF SECTIONS IN THE LAKE
        PRINT 1  ,IN
      1 FORMAT (1H1,27HDEFANT CALCULATIONS FOR THE,2X,I1,2X,12HMODAL SEICH
       1E)
        J =IN/2
        J =2*J
        IF (IN-J)  3,2,3
      2 B =-1.
        GO TO 4
      3 B =1.0
      4 G =35.29728
C       G IS THE ACCELERATION DUE TO GRAVITY AT THE LATITUDE OF THE LAKE
C       THE UNITS OF THIS CONSTANT DETERMINE THE UNITS OF THE OUTPUT
        F =  39.478419/G
        IC =0
C       IC IS A COUNTER FOR THE NUMBER OF ITERATIONS
        K =0
C       THE BOUNDARY CONDITIONS FOLLOW
      5 VD1 =100.0
        HD1 =0.0
        Q1 = 0.0
```

```
      DO 6 I=1,N
      ALPHA = F*DELX(I)/(T3*T3)
      HD2(I) = -(Q1+(VD1+(ALPHA*HD1)/4.0)*V2(I))/(S2(I)+ALPHA*V2(I)/4.0)
      VD2(I) =VD1+ALPHA*(HD1+HD2(I))/2.0
      Q2(I) =Q1+V2(I)*(VD1+VD2(I))/2.0
      Q1 = Q2(I)
      VD1 =VD2(I)
    6 HD1 =HD2(I)
      IC =IC+1
      PRINT 7  ,IC,T3,HD1,VD1,Q1
    7 FORMAT(1H0,2X,9HITERATION  ,I4,5X,5HT3 = ,F11.6,5X,6HHD1 = ,E15.8,5
     1X,6HVD1 = ,E15.8,5X,5HQ1 =  ,E15.8)
      IF (K-1) 8,13,18
    8 A =VD2(N)
      IF (IN-J) 9,10,9
    9 IF (A)  12,23,11
   10 IF (A)  11,23,12
   11 AINCR = -AINCR
   12 K =1
      T3 = T3-AINCR
      GO TO 5
   13 IF (A*VD2(N))  14,23112
   14 IF (AINCR)  15,16,16
   15 T1 = T3+AINCR
      T2 = T3
      GO TO 17
   16 T1 = T3
      T2 = T3+AINCR
```

```
   17  K =2
       GO TO 21
   18  IF (VD2(N)*B)  19,23,22
   19  T2 = T3
   20  IF  (ABS(T3-(T1+T2)/2.0)-0.0001)  23,23,21
   21  T3 =(T1+T2)/2.0
       GO TO 5
   22  T1 = T3
       GO TO 20
C      CONVERT THE PERIOD TO MORE CONVENIENT TIME UNITS
   23  T =T3/60.0
C      WRITE OUT THE RESULTS
       PRINT 24  ,T,(I,HD2(I),VD2(I),Q2(I),I=1,N)
   24  FORMAT(1H1,37HTHE COMPUTED PERIOD OF OSCILLATION IS  ,2X,F10.4,2X,5
      1HHOURS       //(5X,I3,5X,9HHD2(I) = ,E15.7,5X,9HVD2(I) = ,E15.7,5X,8HQ
      22(I) = ,E15.7))
       RETURN
       END
```

APPENDIX 3

```
C       THIS IS A PROGRAM WHICH CALLS SUBROUTINE DEFANT TO COMPUTE THE
C         SEICHE PARAMETERS OF THE FIRST FIVE MODES OF FREE OSCILLATION
C         OF A LAKE OR BAY.
        DIMENSION S2(199),V2(199),DELX(199)
        READ 1,N,(S2(I),V2(I),DELX(I),I=1,N)
     1  FORMAT(I3/(3F10.4))
        SUM1 = 0.0
        SUM2 = 0.0
        SUM3 = 0.0
        SZERO = 0.0
        DO 2 I=1,N
C       FIND THE VOLUME OF THE LAKE BY MULTIPLYING THE AVERAGE CROSS-
C         SECTION AREA TIMES THE DISTANCE BETWEEN SECTIONS.
        SUM1 =SUM1 + (SZERO+S2(I))*0.5*DELX(I)
C       FIND THE LENGTH OF THE TALWEG
        SUM2 =SUM2 + DELX(I)
C       FIND THE TOTAL SURFACE AREA OF THE LAKE
        SUM3 =SUM3 + V2(I)
     2  SZERO = S2(I)
C       ESTIMATE THE UNINODAL PERIOD OF OSCILLATION USING MERIANS FORMULA
        T3 = 2.0*SUM2/SQRT(39.29728*SUM1/SUM3)
        AINCR = 0.1*T3
        DO 3 I=1,5
        CALL DEFANT (S2,V2,N,DELX,T3,AINCR,I)
        AINCR = ABS(AINCR)/FLOAT(I+1)
     3  T3 = T3*FLOAT(I)/FLOAT(I+1)
        STOP
        END
```

APPENDIX 4

```
BEGIN COMMENT ..BURROUGHS EXTENDED ALGOL PROGRAM FOR
   COMPUTING THE SEICHE PARAMETERS OF CLOSED BASINS;
      ARRAY S2,V2,DELX[1:199];
      INTEGER J,I,NODES,N;
      REAL SUM1,SUM2,SUM3,AINCR,SZERO,T3;
      FORMAT FMT1(I2),
            FMT2(3F10.3),
            FMT3(I6,X5,E12.5,X2,E13.6,X4,E14.7,X3,E13.6),
            FMT4(I6,X6,E14.7,X4,E14.7,X4,E14.7),
            FMT5(X1," ITERATION ",X4," PERIOD ",X7," HOR. DISPL.", X6, " VERT.
DISPL. " ,X5," VOL. DISPL. "),
            FMT6(X1," THE COMPUTED PERIOD OF OSCILLATION IS ",E13.6," MI
NUTES."),
            FMT7(X1," SECTION " ,X6," HOR. DISPL. "X7," VERT. DISPL.",X6," VO
L. DISPL.");
      FILE IN READER(1,10);
      FILE OUT ALINE 4(1,10);

     PROCEDURE DEFANT(S2,V2,DELX,N,T3,AINCR,NODES);
      VALUE N,NODES,AINCR;
      INTEGER N,NODES;
      REAL T3,AINCR;
      ARRAY S2,V2,DELX[1];
     BEGIN
      INTEGER IC, K;
      REAL ZALPHA,ETA1,Q1,XE1,G,B,F,A,T1,T2,F1;
      LABEL GUTS,MAELSTROM,HOPE,TESTIT,ISIT,INTER,UPDATE,WHERAMI,NEW,
       OUTPUT;
      ARRAY XE2,Q2,ETA2[1:199];
       NODES : = NODES - 2 x (NODES DIV 2) ; IF NODES = 0 THEN B : = -1.0 ELSE B : = 1.0;
       F : = 4.0 x 3.1415927 x 3.1415927 / 35.29728 ; K: = IC : = 0;
```

```
  WRITE(ALINE[DBL], FMT5);
BEGIN GUTS:  XE1 : = Q1 : = 0.0; ETA1 : = 100.0; F1 : =F / (T3 x T3);
  FOR I : = 1 STEP 1 UNTIL N DO
  BEGIN
   ZALPHA : = F1 x DELX[I];
   XE2[I] : = -(Q1 + (ETA1 + ZALPHA x XE1 x 0.25) x V2[I]) / (S2[I] + ZALPHA x V2[I] x
                                                                                  0.25);
   ETA2[I] : = ETA1 + 0.5 x ZALPHA x (XE1 + XE2[I]);
   Q2[I] : = Q1 + V2[I] x 0.5 x (ETA1 + ETA2[I]);
   XE1 : = XE2[I]; Q1 : = Q2[I]; ETA1 : = ETA2[I]
  END;
 IC : = IC + 1; WRITE(ALINE,FMT3,IC,T3,XE1,ETA1,Q1)
END GUTS;
IF K < 1 THEN  GO TO MAELSTROM
  ELSE IF  K = 1 THEN GO TO HOPE
    ELSE GO TO TESTIT;
MAELSTROM:  A : = XE2[N]; IF NODES =0 THEN GO TO WHERAMI
  ELSE IF A = 0.0 THEN GO TO OUTPUT
    ELSE IF A  < 0.0 THEN GO TO UPDATE
      ELSE INTER: AINCR : = -AINCR; UPDATE : K : = 1; T3 : = T3 - AINCR;
        GO TO GUTS;
WHERAMI: IF A = 0.0 THEN GO TO OUTPUT
    ELSE IF A >0.0 THEN GO TO UPDATE
      ELSE GO TO INTER;
HOPE : F1 : =A x XE2[N]; IF F1 =0.0 THEN GO TO OUTPUT
  ELSE IF F1 <0.0 THEN IF AINCR < 0.0 THEN
            BEGIN T1 : = T3 + AINCR; T2 : = T3; K : =2; GO TO NEW  END
      ELSE BEGIN T1 : = T3; T2 : = T3 + AINCR ; K : = 2; GO TO NEW  END
    ELSE BEGIN T3 : = T3 - AINCR; GO  TO GUTS   END;
TESTIT:  F1 : =B x XE2[N];  IF F1 =0.0 THEN GO TO OUTPUT
```

```
      ELSE IF F1 <0.0 THEN BEGIN T2 : = T3; GO TO ISIT    END
        ELSE T1 : = T3;
    ISIT: IF ABS(T3 -(T1 + T2) x 0.5) ≤ 0.0001 THEN GO TO OUTPUT
      ELSE BEGIN NEW: T3 : =(T1 + T2) x 0.5; GO TO GUTS    END;
    OUTPUT: WRITE(ALINE[DBL], FMT6, T3); WRITE(ALINE[DBL], FMT7);
      FOR I : = 1 STEP 1 UNTIL N DO
        WRITE(ALINE, FMT4, I, XE2[I], ETA2[I], Q2[I]);
   END DEFANT;

  SZERO : = SUM1 : = SUM2 : = SUM3 : = 0.0; READ(READER, FMT1, N);
  FOR I : = 1 STEP 1 UNTIL N DO READ(READER, FMT2, S2[I], V2[I], DELX[I]);
  FOR I : = 1 STEP 1 UNTIL N DO BEGIN SUM1 : =SUM1 + (SZERO + S2[I]) x 0.5 x DELX[I];
    SUM2 : = SUM2 + DELX[I]; SUM3: = SUM3 + V2[I]; SZERO : = S2[I]    END;
  T3 : = 2.0 x SUM2 / SQRT (35.29728 x SUM1 / SUM3); AINCR : =0.1 x T3;
  FOR J : = 1 STEP 1 UNTIL 5 DO BEGIN DEFANT (S2, V2, DELX, N, T3, AINCR, J);
    AINCR : = AINCR / (J + 1); T3: =(J x T3) / (J + 1)    END;
END.
```

ACKNOWLEDGEMENTS

Special thanks go to Dr. R. W. Bachmann, who provided the initial derivation of the Defant equations. Dr. C. H. Mortimer gave invaluable comments and suggestions about the manuscript. Particular thanks go to Miss M. Pierce for document preparation and art work.

REFERENCES

Bachmann, R. W., R. V. Bovbjerg, and J. Hall. 1966. Morphometry of Lake West Okoboji. Iowa Acad. Sci., 73: (in press).

Defant, A. 1918. Neue Methode zur Ermittlung der Eigenschwingungen (Seiches) von abgeschlossenen Wassermassen (Seen, Buchten usw.). Ann. Hydr. Mar. Mit., 46: 78.

_______. 1961. Physical Oceanography, Vol. 2. 598 pp. London: Pergamon Press.

Fee, E. J., and R. W. Bachmann. 1968. An empirical study of the Defant method of seiche analysis. Limnol. Oceanogr. (In press).

Hutchinson, G. E. 1957. A Treatise on Limnology, Vol. I: Geography, Physics, and Chemistry. 1015 pp. New York: John Wiley and Sons.

Johnson, M.A., and E. J. Fee. 1968. A method for computing internal wave phase speeds. (In preparation)

Mortimer, C. H. 1953. The resonant response of stratified lakes to wind. Schweiz. Z. Hydrol., 15: 94-151.

_______. 1965. Spectra of long surface waves and tides in Lake Michigan and Green Bay, Wisconsin. Proc. 8th Conf. Great Lakes Res., Great Lakes Res. Div., Univ. Mich., Publ. No. 13: 304-325.